Chapter 25: Web Services in Visual Basic

Internet & World Wide Web How to Program, 5/e

Note: This chapter is a copy of Chapter 23 of our book *Visual Basic 2010 How to Program*. For that reason, we simply copied the PowerPoint slides for this chapter and *did not* re-numb er them

OBJECTIVES

In this chapter you'll learn:

- How to create WCF web services.
- How XML, JSON, XML-Based Simple Object Access Protocol (SOAP) and Representational State Transfer Architecture (REST) enable WCF web services.
- The elements that comprise WCF web services, such as service references, service endpoints, service contracts and service bindings.
- How to create a client that consumes a WCF web service.
- How to use WCF web services with Windows and web applications.
- How to use session tracking in WCF web services to maintain state information for the client.
- How to pass user-defined types to a WCF web service.

```
1  ' Fig. 23.1: IWelcomeSOAPXMLService.vb
2  ' WCF web service interface that returns a welcome message through SOAP
3  ' protocol and XML format.
4  <ServiceContract()>

5  Public Interface IWelcomeSOAPXMLService
6     ' returns a welcome message
7     <OperationContract()>
8     Function Welcome(ByVal yourName As String) As String
9  End Interface ' IWelcomeSOAPXMLService
```

Fig. 23.1 | WCF web service interface that returns a welcome message through SOAP protocol and XML format.

```
1  ' Fig. 23.2: WelcomeSOAPXMLService.vb
2  ' WCF web service that returns a welcome message through SOAP protocol and
3  ' XML format.
4  Public Class WelcomeSOAPXMLService
5     Implements IWelcomeSOAPXMLService
6
7     ' returns a welcome message
8     Public Function Welcome(ByVal yourName As String) As String _
9        Implements IWelcomeSOAPXMLService.Welcome
10
11       Return "Welcome to WCF Web Services with SOAP and XML, " &
12          yourName & "!"
13    End Function ' Welcome
14 End Class ' WelcomeSOAPXMLService
```

Fig. 23.2 | WCF web service that returns a welcome message through the SOAP protocol and XML format.



Fig. 23.3 | Creating a WCF Service in Visual Web Developer.



Fig. 23.4 | SVC file rendered in a web browser. (Part 1 of 2.)



Fig. 23.4 | SVC file rendered in a web browser. (Part 2 of 2.)



Fig. 23.5 | WCF web service **Properties** window.



Fig. 23.6 | .NET WCF web service client after a web-service reference has been added.



Fig. 23.7 | Interaction between a web-service client and a SOAP web service.

```
1  ' Fig. 23.8: WelcomeSOAPXML.vb
2  ' Client that consumes WelcomeSOAPXMLService.
3  Public Class WelcomeSOAPXML
4     ' reference to web service
5     Private client As New ServiceReference.WelcomeSOAPXMLServiceClient()
6
7     ' creates welcome message from text input and web service
8     Private Sub submitButton_Click(ByVal sender As System.Object,
9        ByVal e As System.EventArgs) Handles submitButton.Click
10
11       MessageBox.Show(client.Welcome(textBox.Text))
12    End Sub ' submitButton_Click
13 End Class ' WelcomeSOAPXML
```

a) User inputs name



Fig. 23.8 | Client that consumes `WelcomeSOAPXMLService`. (Part 1 of 2.)

b) Message sent from **WelcomeSOAPXML-Service**



Fig. 23.8 | Client that consumes `WelcomeSOAPXMLService`. (Part 2 of 2.)

```
1  ' Fig. 23.9: IWelcomeRESTXMLService.vb
2  ' WCF web-service interface. A class that implements this interface
3  ' returns a welcome message through REST architecture and XML data format.
4  Imports System.ServiceModel.Web
5
6  <ServiceContract()>
7  Public Interface IWelcomeRESTXMLService
8     ' returns a welcome message
9     <OperationContract()>
10    <WebGet(UriTemplate:="welcome/{yourName}")>
11    Function Welcome(ByVal yourName As String) As String
12 End Interface ' IWelcomeRESTXMLService
```

Fig. 23.9 | WCF web-service interface. A class that implements this interface returns a welcome message through REST architecture and XML data format.

```
1  ' Fig. 23.10: WelcomeRESTXMLService.vb
2  ' WCF web service that returns a welcome message using REST architecture
3  ' and XML data format.
4  Public Class WelcomeRESTXMLService
5     Implements IWelcomeRESTXMLService
6
7     ' returns a welcome message
8     Public Function Welcome(ByVal yourName As String) _
9        As String Implements IWelcomeRESTXMLService.Welcome
10
11       Return "Welcome to WCF Web Services with REST and XML, " &
12          yourName & "!"
13    End Function ' Welcome
14 End Class ' WelcomeRESTXMLService
```

Fig. 23.10 | WCF web service that returns a welcome message using REST architecture and XML data format.

```
<system.serviceModel>
   <behaviors>
      <serviceBehaviors>
         <behavior>
            <!-- To avoid disclosing metadata information, set the
               value below to false and remove the metadata
               endpoint above before deployment -->
            <serviceMetadata httpGetEnabled="true"/>
            <!-- To receive exception details in faults for debugging
               purposes, set the value below to true.  Set to false
               before deployment to avoid disclosing exception
               information -->
            <serviceDebug includeExceptionDetailInFaults="false"/>
         </behavior>
      </serviceBehaviors>
      <endpointBehaviors>
         <behavior>
            <webHttp/>
         </behavior>
      </endpointBehaviors>
   </behaviors>
   <protocolMapping>
      <add scheme="http" binding="webHttpBinding"/>
   </protocolMapping>
</system.serviceModel>
```

Fig. 23.11 | WelcomeRESTXMLService `Web.config` file.



Fig. 23.12 | Response from `WelcomeRESTXMLService` in XML data format.

```
1  ' Fig. 23.13: WelcomeRESTXML.vb
2  ' Client that consumes the WelcomeRESTXMLService.
3  Imports System.Net
4  Imports System.Xml.Linq
5  Imports <xmlns="http://schemas.microsoft.com/2003/10/Serialization/">
6
7  Public Class WelcomeRESTXML
8     ' object to invoke the WelcomeRESTXMLService
9     Private WithEvents service As New WebClient()
10
11    ' get user input and pass it to the web service
12    Private Sub submitButton_Click(ByVal sender As System.Object,
13       ByVal e As System.EventArgs) Handles submitButton.Click
14
15       ' send request to WelcomeRESTXMLService
16       service.DownloadStringAsync(New Uri(
17          "http://localhost:51424/WelcomeRESTXMLService/Service.svc/" &
18          "welcome/" & textBox.Text))
19    End Sub ' submitButton_Click
20
```

Fig. 23.13 | Client that consumes the `WelcomeRESTXMLService`. (Part 1 of 3.)

```
21      ' process web-service response
22      Private Sub service_DownloadStringCompleted(ByVal sender As Object,
23         ByVal e As System.Net.DownloadStringCompletedEventArgs) _
24         Handles service.DownloadStringCompleted
25
26         ' check if any errors occurred in retrieving service data
27         If e.Error Is Nothing Then
28            ' parse the returned XML string (e.Result)
29            Dim xmlResponse = XDocument.Parse(e.Result)
30
31            ' use XML axis property to access the <string> element's value
32            MessageBox.Show(xmlResponse.<string>.Value)
33         End If
34      End Sub ' service_DownloadStringCompleted
35   End Class ' WelcomeRESTXML
```

Fig. 23.13 | Client that consumes the `WelcomeRESTXMLService`. (Part 2 of 3.)

a) User inputs name.



b) Message sent from **WelcomeRESTXMLService**.



Fig. 23.13 | Client that consumes the `WelcomeRESTXMLService`. (Part 3 of 3.)

```
1  ' Fig. 23.14: IWelcomeRESTJSONService.vb
2  ' WCF web-service interface that returns a welcome message through REST
3  ' architecture and JSON format.
4  Imports System.ServiceModel.Web
5
6  <ServiceContract()>
7  Public Interface IWelcomeRESTJSONService
```

Fig. 23.14 | WCF web-service interface that returns a welcome message through REST architecture and JSON format. (Part 1 of 2.)

```
8       ' returns a welcome message
9       <OperationContract()>
10      <WebGet(ResponseFormat:=WebMessageFormat.Json,
11         UriTemplate:="welcome/{yourName}")>
12      Function Welcome(ByVal yourName As String) As TextMessage
13   End Interface ' IWelcomeRESTJSONService
14
15   ' class to encapsulate a String to send in JSON format
16   <DataContract()>
17   Public Class TextMessage
18      Public messageValue As String
19
20      ' property Message
21      <DataMember()>
22      Public Property Message() As String
23         Get
24            Return messageValue
25         End Get
26         Set(ByVal value As String)
27            messageValue = value
28         End Set
29      End Property ' Message
30   End Class ' TextMessage
```

Fig. 23.14 | WCF web-service interface that returns a welcome message through REST architecture and JSON format. (Part 2 of 2.)

```vb
' Fig. 23.15: WelcomeRESTJSONService.vb
' WCF web service that returns a welcome message through REST architecture
' and JSON format.
Public Class WelcomeRESTJSONService
   Implements IWelcomeRESTJSONService

   ' returns a welcome message
   Public Function Welcome(ByVal yourName As String)
      As TextMessage Implements IWelcomeRESTJSONService.Welcome
      ' add welcome message to field of TextMessage object
      Dim welcomeString As New TextMessage
      welcomeString.Message = "Welcome to WCF Web Services with REST " &
         "and JSON, " & yourName & "!"
      Return welcomeString
   End Function ' Welcome
End Class ' WelcomeRESTJSONService
```

Fig. 23.15 | WCF web service that returns a welcome message through REST architecture and JSON format.



Fig. 23.16 | Response from `WelcomeRESTJSONService` in JSON data format.

```
1  ' Fig. 23.17: WelcomeRESTJSON.vb
2  ' Client that consumes WelcomeRESTJSONService.
3  Imports System.IO
4  Imports System.Net
5  Imports System.Runtime.Serialization.Json
6  Imports System.Text
7
8  Public Class WelcomeRESTJSON
9     ' object to invoke the WelcomeRESTJSONService
10    Private WithEvents service As New WebClient()
11
12    ' creates welcome message from text input and web service
13    Private Sub submitButton_Click(ByVal sender As System.Object,
14       ByVal e As System.EventArgs) Handles submitButton.Click
15
16       ' send request to WelcomeRESTJSONService
17       service.DownloadStringAsync(New Uri(
18          "http://localhost:49745/WelcomeRESTJSONService/Service.svc/" &
19        "welcome/" & textBox.Text))
20    End Sub ' submitButton
21
```

Fig. 23.17 | Client that consumes `WelcomeRESTJSONService`. (Part 1 of 3.)

```
22      ' process web-service response
23      Private Sub service_DownloadStringCompleted(ByVal sender As Object,
24         ByVal e As System.Net.DownloadStringCompletedEventArgs) _
25         Handles service.DownloadStringCompleted
26
27         ' check if any errors occurred in retrieving service data
28         If e.Error Is Nothing Then
29            ' deserialize response into a TextMessage object
30            Dim JSONSerializer _
31               As New DataContractJsonSerializer(GetType(TextMessage))
32            Dim welcomeString = JSONSerializer.ReadObject(
33               New MemoryStream(Encoding.Unicode.GetBytes(e.Result)))
34
35            ' display Message text
36            MessageBox.Show(CType(welcomeString, TextMessage).Message)
37         End If
38      End Sub ' service_DownloadStringCompleted
39   End Class ' WelcomeRESTJSON
40
```

Fig. 23.17 | Client that consumes `WelcomeRESTJSONService`. (Part 2 of 3.)

```
41  ' TextMessage class representing a JSON object
42  <Serializable()>
43  Public Class TextMessage
44     Public Message As String
45  End Class ' TextMessage
```

a) User inputs name.



b) Message sent from `WelcomeRESTJSONService`.



Fig. 23.17 | Client that consumes `WelcomeRESTJSONService`. (Part 3 of 3.)

```
1  ' Fig. 23.18: IBlackjackService.vb
2  ' Blackjack game WCF web-service interface.
3  <ServiceContract(SessionMode:=SessionMode.Required)> _
4  Public Interface IBlackjackService
5     ' deals a card that has not been dealt
6     <OperationContract()>
7     Function DealCard() As String
8
9     ' creates and shuffles the deck
10    <OperationContract()>
11    Sub Shuffle()
12
13    ' calculates value of a hand
14    <OperationContract()>
15    Function GetHandValue(ByVal dealt As String) As Integer
16 End Interface ' IBlackjackService
```

Fig. 23.18 | Blackjack game WCF web-service interface.

```
1  ' Fig. 23.19: BlackjackService.vb
2  ' Blackjack game WCF web service.
3  Imports System.Collections.Generic
4
5  <ServiceBehavior(InstanceContextMode:=InstanceContextMode.PerSession)>
6  Public Class BlackjackService
7     Implements IBlackjackService
8     ' create persistent session deck-of-cards object
9     Dim deck As New List(Of String)
10
11    ' deals card that has not yet been dealt
12    Public Function DealCard() As String _
13       Implements IBlackjackService.DealCard
14
15       Dim card As String = Convert.ToString(deck(0)) ' get first card
16       deck.RemoveAt(0) ' remove card from deck
17       Return card
18    End Function ' DealCard
19
```

Fig. 23.19 | Blackjack game WCF web service. (Part 1 of 4.)

```
20      ' creates and shuffles a deck of cards
21      Public Sub Shuffle() Implements IBlackjackService.Shuffle
22         Dim randomObject As New Random() ' generates random numbers
23
24         deck.Clear() ' clears deck for new game
25
26         ' generate all possible cards
27         For face = 1 To 13 ' loop through face values
28            For suit As Integer = 0 To 3 ' loop through suits
29               deck.Add(face & " " & suit) ' add card (string) to deck
30            Next suit
31         Next face
32
33         ' shuffles deck by swapping each card with another card randomly
34         For i = 0 To deck.Count - 1
35            ' get random index
36            Dim newIndex = randomObject.Next(deck.Count - 1)
37            Dim temporary = deck(i) ' save current card in temporary variable
38            deck(i) = deck(newIndex) ' copy randomly selected card
39            deck(newIndex) = temporary ' copy current card back into deck
40         Next
41      End Sub ' Shuffle
42
```

Fig. 23.19 | Blackjack game WCF web service. (Part 2 of 4.)

```
43      ' computes value of hand
44      Public Function GetHandValue(ByVal dealt As String) As Integer _
45         Implements IBlackjackService.GetHandValue
46         ' split string containing all cards
47         Dim tab As Char = Convert.ToChar(vbTab)
48         Dim cards As String() = dealt.Split(tab) ' get array of cards
49         Dim total As Integer = 0 ' total value of cards in hand
50         Dim face As Integer ' face of the current card
51         Dim aceCount As Integer = 0 ' number of aces in hand
52
53         ' loop through the cards in the hand
54         For Each card In cards
55            ' get face of card
56            face = Convert.ToInt32(card.Substring(0, card.IndexOf(" ")))
57
58            Select Case face
59               Case 1 ' if ace, increment aceCount
60                  aceCount += 1
61               Case 11 To 13 ' if jack, queen or king add 10
62                  total += 10
63               Case Else ' otherwise, add value of face
64                  total += face
65            End Select
66         Next
67
```

Fig. 23.19 | Blackjack game WCF web service. (Part 3 of 4.)

```
68          ' if there are any aces, calculate optimum total
69          If aceCount > 0 Then
70             ' if it is possible to count one ace as 11, and the rest
71             ' as 1 each, do so; otherwise, count all aces as 1 each
72             If (total + 11 + aceCount - 1 <= 21) Then
73                total += 11 + aceCount - 1
74             Else
75                total += aceCount
76             End If
77          End If
78
79          Return total
80       End Function ' GetHandValue
81    End Class ' BlackjackService
```

Fig. 23.19 | Blackjack game WCF web service. (Part 4 of 4.)

```
1  ' Fig. 23.20: Blackjack.vb
2  ' Blackjack game that uses the BlackjackService web service.
3  Imports System.Net
4
5  Public Class Blackjack
6     ' reference to web service
7     Private dealer As ServiceReference.BlackJackServiceClient
8
9     ' string representing the dealer's cards
10    Private dealersCards As String
11
12    ' string representing the player's cards
13    Private playersCards As String
14    Private cardBoxes As List(Of PictureBox) ' list of card images
15    Private currentPlayerCard As Integer ' player's current card number
16    Private currentDealerCard As Integer ' dealer's current card number
17
```

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 1 of 18.)

```
18      ' enum representing the possible game outcomes
19      Public Enum GameStatus
20         PUSH ' game ends in a tie
21         LOSE ' player loses
22         WIN ' player wins
23         BLACKJACK ' player has blackjack
24      End Enum ' GameStatus
25
26      ' sets up the game
27      Private Sub Blackjack_Load(ByVal sender As Object,
28         ByVal e As System.EventArgs) Handles Me.Load
29         ' instantiate object allowing communication with web service
30         dealer = New ServiceReference.BlackJackServiceClient()
31
32         cardBoxes = New List(Of PictureBox)
33
34         ' put PictureBoxes into cardBoxes List
35         cardBoxes.Add(pictureBox1)
36         cardBoxes.Add(pictureBox2)
37         cardBoxes.Add(pictureBox3)
38         cardBoxes.Add(pictureBox4)
39         cardBoxes.Add(pictureBox5)
40         cardBoxes.Add(pictureBox6)
```

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 2 of 18.)

```
41          cardBoxes.Add(pictureBox7)
42          cardBoxes.Add(pictureBox8)
43          cardBoxes.Add(pictureBox9)
44          cardBoxes.Add(pictureBox10)
45          cardBoxes.Add(pictureBox11)
46          cardBoxes.Add(pictureBox12)
47          cardBoxes.Add(pictureBox13)
48          cardBoxes.Add(pictureBox14)
49          cardBoxes.Add(pictureBox15)
50          cardBoxes.Add(pictureBox16)
51          cardBoxes.Add(pictureBox17)
52          cardBoxes.Add(pictureBox18)
53          cardBoxes.Add(pictureBox19)
54          cardBoxes.Add(pictureBox20)
55          cardBoxes.Add(pictureBox21)
56          cardBoxes.Add(pictureBox22)
57      End Sub ' Blackjack_Load
58
```

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 3 of 18.)

```
59      ' deals cards to dealer while dealer's total is less than 17,
60      ' then computes value of each hand and determines winner
61      Private Sub DealerPlay()
62         ' reveal dealer's second card
63         Dim tab As Char = Convert.ToChar(vbTab)
64         Dim cards As String() = dealersCards.Split(tab)
65         DisplayCard(1, cards(1))
66
67         Dim nextCard As String
68
69         ' while value of dealer's hand is below 17,
70         ' dealer must take cards
71         While dealer.GetHandValue(dealersCards) < 17
72            nextCard = dealer.DealCard() ' deal new card
73            dealersCards &= vbTab & nextCard
74
75            ' update GUI to show new card
76            MessageBox.Show("Dealer takes a card")
77            DisplayCard(currentDealerCard, nextCard)
78            currentDealerCard += 1
79         End While
80
```

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 4 of 18.)

```
81          Dim dealerTotal As Integer = dealer.GetHandValue(dealersCards)
82          Dim playerTotal As Integer = dealer.GetHandValue(playersCards)
83
84          ' if dealer busted, player wins
85          If dealerTotal > 21 Then
86             GameOver(GameStatus.WIN)
87          Else
88             ' if dealer and player have not exceeded 21,
89             ' higher score wins; equal scores is a push.
90             If dealerTotal > playerTotal Then ' player loses game
91                GameOver(GameStatus.LOSE)
92             ElseIf playerTotal > dealerTotal Then ' player wins game
93                GameOver(GameStatus.WIN)
94             Else ' player and dealer tie
95                GameOver(GameStatus.PUSH)
96             End If
97          End If
98       End Sub ' DealerPlay
99
```

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 5 of 18.)

```
100     ' displays card represented by cardValue in specified PictureBox
101     Public Sub DisplayCard(
102        ByVal card As Integer, ByVal cardValue As String)
103        ' retrieve appropriate PictureBox
104        Dim displayBox As PictureBox = cardBoxes(card)
105
106        ' if string representing card is empty,
107        ' set displayBox to display back of card
108        If String.IsNullOrEmpty(cardValue) Then
109           displayBox.Image =
110              Image.FromFile("blackjack_images/cardback.png")
111           Return
112        End If
113
114        ' retrieve face value of card from cardValue
115        Dim face As String =
116           cardValue.Substring(0, cardValue.IndexOf(" "))
117
118        ' retrieve the suit of the card from cardValue
119        Dim suit As String =
120           cardValue.Substring(cardValue.IndexOf(" ") + 1)
121
122        Dim suitLetter As Char ' suit letter used to form image file name
```

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 6 of 18.)

```
123
124         ' determine the suit letter of the card
125         Select Case Convert.ToInt32(suit)
126            Case 0 ' clubs
127               suitLetter = "c"c
128            Case 1 ' diamonds
129               suitLetter = "d"c
130            Case 2 ' hearts
131               suitLetter = "h"c
132            Case Else ' spades
133               suitLetter = "s"c
134         End Select
135
136         ' set displayBox to display appropriate image
137         displayBox.Image = Image.FromFile(
138            "blackjack_images/" & face & suitLetter & ".png")
139      End Sub ' DisplayCard
140
```

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 7 of 18.)

```
141     ' displays all player cards and shows
142     ' appropriate game status message
143     Public Sub GameOver(ByVal winner As GameStatus)
144        ' display appropriate status image
145        If winner = GameStatus.PUSH Then ' push
146           statusPictureBox.Image =
147              Image.FromFile("blackjack_images/tie.png")
148        ElseIf winner = GameStatus.LOSE Then ' player loses
149           statusPictureBox.Image =
150              Image.FromFile("blackjack_images/lose.png")
151        ElseIf winner = GameStatus.BLACKJACK Then
152           ' player has blackjack
153           statusPictureBox.Image =
154              Image.FromFile("blackjack_images/blackjack.png")
155        Else ' player wins
156           statusPictureBox.Image =
157              Image.FromFile("blackjack_images/win.png")
158        End If
159
160        ' display final totals for dealer and player
161        dealerTotalLabel.Text =
162           "Dealer: " & dealer.GetHandValue(dealersCards)
```

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 8 of 18.)

```
163         playerTotalLabel.Text =
164            "Player: " & dealer.GetHandValue(playersCards)
165
166         ' reset controls for new game
167         stayButton.Enabled = False
168         hitButton.Enabled = False
169         dealButton.Enabled = True
170      End Sub ' GameOver
171
172      ' deal two cards each to dealer and player
173      Private Sub dealButton_Click(ByVal sender As System.Object,
174         ByVal e As System.EventArgs) Handles dealButton.Click
175         Dim card As String ' stores a card temporarily until added to a hand
176
177         ' clear card images
178         For Each cardImage As PictureBox In cardBoxes
179            cardImage.Image = Nothing
180         Next
181
182         statusPictureBox.Image = Nothing ' clear status image
183         dealerTotalLabel.Text = String.Empty ' clear final total for dealer
184         playerTotalLabel.Text = String.Empty ' clear final total for player
185
```

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 9 of 18.)

```
        ' create a new, shuffled deck on the web service host
        dealer.Shuffle()

        ' deal two cards to player
        playersCards = dealer.DealCard() ' deal a card to player's hand

        ' update GUI to display new card
        DisplayCard(11, playersCards)
        card = dealer.DealCard() ' deal a second card
        DisplayCard(12, card) ' update GUI to display new card
        playersCards &= vbTab & card ' add second card to player's hand

        ' deal two cards to dealer, only display face of first card
        dealersCards = dealer.DealCard() ' deal a card to dealer's hand
        DisplayCard(0, dealersCards) ' update GUI to display new card
        card = dealer.DealCard() ' deal a second card
        DisplayCard(1, String.Empty) ' update GUI to show face-down card
        dealersCards &= vbTab & card ' add second card to dealer's hand

        stayButton.Enabled = True ' allow player to stay
        hitButton.Enabled = True ' allow player to hit
        dealButton.Enabled = False ' disable Deal Button

```

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 10 of 18.)

```
209         ' determine the value of the two hands
210         Dim dealerTotal As Integer = dealer.GetHandValue(dealersCards)
211         Dim playerTotal As Integer = dealer.GetHandValue(playersCards)
212
213         ' if hands equal 21, it is a push
214         If dealerTotal = playerTotal And dealerTotal = 21 Then
215            GameOver(GameStatus.PUSH)
216         ElseIf dealerTotal = 21 Then ' if dealer has 21, dealer wins
217            GameOver(GameStatus.LOSE)
218         ElseIf playerTotal = 21 Then ' player has blackjack
219            GameOver(GameStatus.BLACKJACK)
220         End If
221
222         currentDealerCard = 2 ' next dealer card has index 2 in cardBoxes
223         currentPlayerCard = 13 ' next player card has index 13 in cardBoxes
224      End Sub ' dealButton_Click
225
```

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 11 of 18.)

```
226     ' deal another card to player
227     Private Sub hitButton_Click(ByVal sender As System.Object,
228        ByVal e As System.EventArgs) Handles hitButton.Click
229        ' get player another card
230        Dim card As String = dealer.DealCard() ' deal new card
231        playersCards &= vbTab & card ' add new card to player's hand
232
233        ' update GUI to show new card
234        DisplayCard(currentPlayerCard, card)
235        currentPlayerCard += 1
236
237        ' determine the value of the player's hand
238        Dim total As Integer = dealer.GetHandValue(playersCards)
239
240        ' if player exceeds 21, house wins
241        If total > 21 Then
242           GameOver(GameStatus.LOSE)
243        End If
244
245        ' if player has 21,
246        ' they cannot take more cards, and dealer plays
247        If total = 21 Then
248           hitButton.Enabled = False
```

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 12 of 18.)

```
            DealerPlay()
         End If
      End Sub ' hitButton_Click

      ' play the dealer's hand after the play chooses to stay
      Private Sub stayButton_Click(ByVal sender As System.Object,
         ByVal e As System.EventArgs) Handles stayButton.Click
         stayButton.Enabled = False ' disable Stay Button
         hitButton.Enabled = False ' disable Hit Button
         dealButton.Enabled = True ' re-enable Deal Button
         DealerPlay() ' player chose to stay, so play the dealer's hand
      End Sub ' stayButton_Click
   End Class ' Blackjack
```

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 13 of 18.)

a) Initial cards dealt to the player and the dealer when the user presses the **Deal** button.

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 14 of 18.)

b) Cards after the player presses the **Hit** button once, then the **Stay** button. In this case, the player wins the game with a higher total than the dealer.

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 15 of 18.)

c) Cards after the player presses the **Hit** button once, then the **Stay** button. In this case, the player busts (exceeds 21) and the dealer wins the game.



Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 16 of 18.)

d) Cards after the player presses the **Deal** button. In this case, the player wins with Blackjack because the first two cards are an ace and a card with a value of 10 (a jack in this case).

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 17 of 18.)

e) Cards after the player presses the **Stay** button. In this case, the player and dealer push—they have the same card total.

Fig. 23.20 | Blackjack game that uses the `BlackjackService` web service. (Part 18 of 18.)

```
1  ' Fig. 23.21: IReservationService.vb
2  ' Airline reservation WCF web-service interface.
3  <ServiceContract()>
4  Public Interface IReservationService
5     ' reserves a seat
6     <OperationContract()>
7     Function Reserve(ByVal seatType As String,
8        ByVal classType As String) As Boolean
9  End Interface ' IReservationService
```

Fig. 23.21 | Airline reservation WCF web-service interface.

```vb
' Fig. 23.22: ReservationService.vb
' Airline reservation WCF web service.
Public Class ReservationService
   Implements IReservationService

   ' create ticketsDB object to access Tickets database
   Private ticketsDB As New TicketsDataContext()

   ' checks database to determine whether matching seat is available
   Public Function Reserve(ByVal seatType As String,
      ByVal classType As String) As Boolean _
      Implements IReservationService.Reserve

      ' LINQ query to find seats matching the parameters
      Dim result =
         From seat In ticketsDB.Seats
         Where (seat.Taken = 0) And (seat.SeatType = seatType)
            And (seat.SeatClass = classType)

```

Fig. 23.22 | Airline reservation WCF web service. (Part 1 of 2.)

```
        ' if the number of seats returned is nonzero,
        ' obtain the first matching seat number and mark it as taken
        If result.Count() <> 0 Then
           ' get first available seat
           Dim firstAvailableSeat As Seat = result.First()
           firstAvailableSeat.Taken = 1 ' mark the seat as taken
           ticketsDB.SubmitChanges() ' update
           Return True ' seat was reserved
        End If

        Return False ' no seat was reserved
     End Function ' Reserve
  End Class ' ReservationService
```

Fig. 23.22 | Airline reservation WCF web service. (Part 2 of 2.)



Fig. 23.23 | ASPX file that takes reservation information.

```vb
' Fig. 23.24: ReservationClient.aspx.vb
' ReservationClient code-behind file.
Partial Class ReservationClient
   Inherits System.Web.UI.Page
   ' object of proxy type used to connect to Reservation service
   Private ticketAgent As New ServiceReference.ReservationServiceClient()

   Protected Sub reserveButton_Click(ByVal sender As Object,
      ByVal e As System.EventArgs) Handles reserveButton.Click

      ' if the ticket is reserved
      If ticketAgent.Reserve(seatList.SelectedItem.Text,
         classList.SelectedItem.Text.ToString()) Then

         ' hide other controls
         instructionsLabel.Visible = False
         seatList.Visible = False
         classList.Visible = False
         reserveButton.Visible = False
         errorLabel.Visible = False

         ' display message indicating success
         Response.Write("Your reservation has been made. Thank you.")
```

Fig. 23.24 | ReservationClient code-behind file. (Part 1 of 2.)

```
24          Else ' service method returned false, so signal failure
25             ' display message in the initially blank errorLabel
26             errorLabel.Text = "This type of seat is not available. " &
27                "Please modify your request and try again."
28          End If
29       End Sub ' reserveButton_Click
30    End Class ' ReservationClient
```

Fig. 23.24 | `ReservationClient` code-behind file. (Part 2 of 2.)

a) Selecting a seat.



b) Seat is reserved successfully.



Fig. 23.25 | Ticket reservation web-application sample execution. (Part 1 of 2.)

c) Attempting to reserve another seat.



d) No seats match the requested type and class.



Fig. 23.25 | Ticket reservation web-application sample execution. (Part 2 of 2.)

```
1  ' Fig. 23.26: Equation.vb
2  ' Class Equation that contains information about an equation.
3  <DataContract()>
4  Public Class Equation
5     Private leftOperand As Integer ' number to the left of the operator
6     Private rightOperand As Integer ' number to the right of the operator
7     Private resultValue As Integer ' result of the operation
8     Private operationType As String ' type of the operation
9
10    ' required default constructor
11    Public Sub New()
12       MyClass.New(0, 0, "add")
13    End Sub ' parameterless New
14
```

Fig. 23.26 | Class `Equation` that contains information about an equation. (Part 1 of 6.)

```
15      ' three-argument constructor for class Equation
16      Public Sub New(ByVal leftValue As Integer,
17         ByVal rightValue As Integer, ByVal type As String)
18
19         Left = leftValue
20         Right = rightValue
21
22         Select Case type ' perform appropriate operation
23            Case "add" ' addition
24               Result = leftOperand + rightOperand
25               operationType = "+"
26            Case "subtract" ' subtraction
27               Result = leftOperand - rightOperand
28               operationType = "-"
29            Case "multiply" ' multiplication
30               Result = leftOperand * rightOperand
31               operationType = "*"
32         End Select
33      End Sub ' three-parameter New
34
```

Fig. 23.26 | Class `Equation` that contains information about an equation. (Part 2 of 6.)

```
35      ' return string representation of the Equation object
36      Public Overrides Function ToString() As String
37         Return leftOperand.ToString() & " " & operationType & " " &
38            rightOperand.ToString() & " = " & resultValue.ToString()
39      End Function ' ToString
40
41      ' property that returns a string representing left-hand side
42      <DataMember()>
43      Public Property LeftHandSide() As String
44         Get
45            Return leftOperand.ToString() & " " & operationType & " " &
46               rightOperand.ToString()
47         End Get
48
49         Set(ByVal value As String) ' required set accessor
50            ' empty body
51         End Set
52      End Property ' LeftHandSide
53
```

Fig. 23.26 | Class `Equation` that contains information about an equation. (Part 3 of 6.)

```
54      ' property that returns a string representing right-hand side
55      <DataMember()>
56      Public Property RightHandSide() As String
57         Get
58            Return resultValue.ToString()
59         End Get
60
61         Set(ByVal value As String) ' required set accessor
62            ' empty body
63         End Set
64      End Property ' RightHandSide
65
66      ' property to access the left operand
67      <DataMember()>
68      Public Property Left() As Integer
69         Get
70            Return leftOperand
71         End Get
72
73         Set(ByVal value As Integer)
74            leftOperand = value
75         End Set
76      End Property ' Left
77
```

Fig. 23.26 | Class `Equation` that contains information about an equation. (Part 4 of 6.)

```
78      ' property to access the right operand
79      <DataMember()>
80      Public Property Right() As Integer
81         Get
82            Return rightOperand
83         End Get
84
85         Set(ByVal value As Integer)
86            rightOperand = value
87         End Set
88      End Property ' Right
89
90      ' property to access the result of applying
91      ' an operation to the left and right operands
92      <DataMember()>
93      Public Property Result() As Integer
94         Get
95            Return resultValue
96         End Get
97
98         Set(ByVal value As Integer)
99            resultValue = value
100        End Set
101     End Property ' Result
```

Fig. 23.26 | Class `Equation` that contains information about an equation. (Part 5 of 6.)

```
102
103     ' property to access the operation
104     <DataMember()>
105     Public Property Operation() As String
106        Get
107           Return operationType
108        End Get
109
110        Set(ByVal value As String)
111           operationType = value
112        End Set
113     End Property ' Operation
114  End Class ' Equation
```

Fig. 23.26 | Class `Equation` that contains information about an equation. (Part 6 of 6.)

```
1  ' Fig. 23.27: IEquationGeneratorService.vb
2  ' WCF REST service interface to create random equations based on a
3  ' specified operation and difficulty level.
4  Imports System.ServiceModel.Web
5
6  <ServiceContract()>
7  Public Interface IEquationGeneratorService
8     ' method to generate a math equation
9     <OperationContract()>
10    <WebGet(UriTemplate:="equation/{operation}/{level}")>
11    Function GenerateEquation(ByVal operation As String,
12       ByVal level As String) As Equation
13 End Interface ' IEquationGeneratorService
```

Fig. 23.27 | WCF REST service interface to create random equations based on a specified operation and difficulty level.

```vb
' Fig. 23.28: EquationGeneratorService.vb
' WCF REST service to create random equations based on a
' specified operation and difficulty level.
Public Class EquationGeneratorService
   Implements IEquationGeneratorService
   ' method to generate a math equation
   Public Function GenerateEquation(ByVal operation As String,
      ByVal level As String) As Equation _
      Implements IEquationGeneratorService.GenerateEquation

      ' convert level from String to Integer
      Dim digits = Convert.ToInt32(level)

      ' calculate maximum and minimum number to be used
      Dim maximum As Integer = Convert.ToInt32(Math.Pow(10, digits))
      Dim minimum As Integer = Convert.ToInt32(Math.Pow(10, digits - 1))

      Dim randomObject As New Random() ' used to generate random numbers

```

Fig. 23.28 | WCF REST service to create random equations based on a specified operation and difficulty level. (Part 1 of 2.)

```
20          ' create Equation consisting of two random
21          ' numbers in the range minimum to maximum
22          Dim newEquation As New Equation(
23             randomObject.Next(minimum, maximum),
24             randomObject.Next(minimum, maximum), operation)
25
26          Return newEquation
27       End Function ' GenerateEquation
28    End Class ' EquationGeneratorService
```

Fig. 23.28 | WCF REST service to create random equations based on a specified operation and difficulty level. (Part 2 of 2.)

```
1  ' Fig. 23.29: MathTutor.vb
2  ' Math tutor using EquationGeneratorService to create equations.
3  Imports System.Net
4  Imports System.Xml.Linq
5  Imports <xmlns="http://schemas.datacontract.org/2004/07/">
6
7  Public Class MathTutor
8     Private operation As String = "add" ' the default operation
9     Private level As Integer = 1 ' the default difficulty level
10    Private leftHandSide As String ' the left side of the equation
11    Private result As Integer ' the answer
12
13    Private WithEvents service As New WebClient() ' used to invoke service
14
15    ' generates a new equation when user clicks button
16    Private Sub generateButton_Click(ByVal sender As System.Object,
17       ByVal e As System.EventArgs) Handles generateButton.Click
18
19       ' send request to EquationGeneratorServiceXML
20       service.DownloadStringAsync(New Uri(
21          "http://localhost:49593/EquationGeneratorServiceXML/" &
22          "Service.svc/equation/" & operation & "/" & level))
23    End Sub ' generateButton_Click
```

Fig. 23.29 | Math tutor using XML version of EquationGeneratorService to create equations. (Part 1 of 7.)

```
24
25      ' process web-service response
26      Private Sub service_DownloadStringCompleted(ByVal sender As Object,
27         ByVal e As System.Net.DownloadStringCompletedEventArgs) _
28         Handles service.DownloadStringCompleted
29
30         ' check if any errors occurred in retrieving service data
31         If e.Error Is Nothing Then
32            ' parse response and get LeftHandSide and Result values
33            Dim xmlResponse = XDocument.Parse(e.Result)
34            leftHandSide = xmlResponse.<Equation>.<LeftHandSide>.Value
35            result = Convert.ToInt32(xmlResponse.<Equation>.<Result>.Value)
36
37            questionLabel.Text = leftHandSide ' display left side of equation
38            okButton.Enabled = True ' enable okButton
39            answerTextBox.Enabled = True ' enable answerTextBox
40         End If
41      End Sub ' service_DownloadStringCompleted
42
```

Fig. 23.29 | Math tutor using XML version of `EquationGeneratorService` to create equations. (Part 2 of 7.)

```
43      ' check user's answer
44      Private Sub okButton_Click(ByVal sender As System.Object,
45         ByVal e As System.EventArgs) Handles okButton.Click
46
47         If Not String.IsNullOrEmpty(answerTextBox.Text) Then
48            ' get user's answer
49            Dim userAnswer As Integer = Convert.ToInt32(answerTextBox.Text)
50
51            ' determine whether user's answer is correct
52            If result = userAnswer Then
53               questionLabel.Text = String.Empty ' clear question
54               answerTextBox.Clear() ' clear answer
55               okButton.Enabled = False ' disable OK button
56               MessageBox.Show("Correct! Good job!")
57            Else
58               MessageBox.Show("Incorrect. Try again.")
59            End If
60         End If
61      End Sub ' okButton_Click
62
```

Fig. 23.29 | Math tutor using XML version of `EquationGeneratorService` to create equations. (Part 3 of 7.)

```
63      ' set the operation to addition
64      Private Sub additionRadioButton_CheckedChanged(
65         ByVal sender As System.Object, ByVal e As System.EventArgs) _
66         Handles additionRadioButton.CheckedChanged
67         operation = "add"
68      End Sub ' additionRadioButton_CheckedChanged
69
70      ' set the operation to subtraction
71      Private Sub subtractionRadioButton_CheckedChanged(
72         ByVal sender As System.Object, ByVal e As System.EventArgs) _
73         Handles subtractionRadioButton.CheckedChanged
74         operation = "subtract"
75      End Sub ' subtractionRadioButton_CheckedChanged
76
77      ' set the operation to multiplication
78      Private Sub multiplicationRadioButton_CheckedChanged(
79         ByVal sender As System.Object, ByVal e As System.EventArgs) _
80         Handles multiplicationRadioButton.CheckedChanged
81         operation = "multiply"
82      End Sub ' multiplicationRadioButton_CheckedChanged
83
```

Fig. 23.29 | Math tutor using XML version of `EquationGeneratorService` to create equations. (Part 4 of 7.)

```
84      ' set difficulty level to 1
85      Private Sub levelOneRadioButton_CheckedChanged(
86         ByVal sender As System.Object, ByVal e As System.EventArgs) _
87         Handles levelOneRadioButton.CheckedChanged
88         level = 1
89      End Sub ' levelOneRadioButton_CheckedChanged
90
91      ' set difficulty level to 2
92      Private Sub levelTwoRadioButton_CheckedChanged(
93         ByVal sender As System.Object, ByVal e As System.EventArgs) _
94         Handles levelTwoRadioButton.CheckedChanged
95         level = 2
96      End Sub ' levelTwoRadioButton_CheckedChanged
97
98      ' set difficulty level to 3
99      Private Sub levelThreeRadioButton_CheckedChanged(
100        ByVal sender As System.Object, ByVal e As System.EventArgs) _
101        Handles levelThreeRadioButton.CheckedChanged
102        level = 3
103     End Sub ' levelThreeRadioButton_CheckedChanged
104  End Class ' MathTutor
```

Fig. 23.29 | Math tutor using XML version of `EquationGeneratorService` to create equations. (Part 5 of 7.)

a) Generating a level 1 addition equation.



b) Answering the question incorrectly.





Fig. 23.29 | Math tutor using XML version of `EquationGeneratorService` to create equations. (Part 6 of 7.)

c) Answering the question correctly.





Fig. 23.29 | Math tutor using XML version of `EquationGeneratorService` to create equations. (Part 7 of 7.)

```
1  ' Fig. 23.30: IEquationGeneratorService.vb
2  ' WCF REST service interface to create random equations based on a
3  ' specified operation and difficulty level.
4  Imports System.ServiceModel.Web
5
6  <ServiceContract()>
7  Public Interface IEquationGeneratorService
8     ' method to generate a math equation
9     <OperationContract()>
10    <WebGet(ResponseFormat:=WebMessageFormat.Json,
11       UriTemplate:="equation/{operation}/{level}")>
12    Function GenerateEquation(ByVal operation As String,
13       ByVal level As String) As Equation
14 End Interface ' IEquationGeneratorService
```

Fig. 23.30 | WCF REST service interface to create random equations based on a specified operation and difficulty level.

```
1  ' Fig. 23.31: MathTutor.vb
2  ' Math tutor using EquationGeneratorServiceJSON to create equations.
3  Imports System.Net
4  Imports System.IO
5  Imports System.Text
6  Imports System.Runtime.Serialization.Json
7
8  Public Class MathTutor
9     Private operation As String = "add" ' the default operation
10    Private level As Integer = 1 ' the default difficulty level
11    Private currentEquation As Equation ' represents the Equation
12    Private WithEvents service As New WebClient() ' used to invoke service
13
14    ' generates a new equation when user clicks button
15    Private Sub generateButton_Click(ByVal sender As System.Object,
16       ByVal e As System.EventArgs) Handles generateButton.Click
17
18       ' send request to EquationGeneratorServiceJSON
19       service.DownloadStringAsync(New Uri(
20          "http://localhost:49817/EquationGeneratorServiceJSON/" &
21          "Service.svc/equation/" & operation & "/" & level))
22    End Sub ' generateButton_Click
```

Fig. 23.31 | Math tutor using JSON version of EquationGeneratorServiceJSON. (Part 1 of 7.)

```
23
24      ' process web-service response
25      Private Sub service_DownloadStringCompleted(ByVal sender As Object,
26         ByVal e As System.Net.DownloadStringCompletedEventArgs) _
27         Handles service.DownloadStringCompleted
28
29         ' check if any errors occurred in retrieving service data
30         If e.Error Is Nothing Then
31            ' deserialize response into an equation object
32            Dim JSONSerializer As New
33               DataContractJsonSerializer(GetType(Equation))
34            currentEquation = CType(JSONSerializer.ReadObject(New
35               MemoryStream(Encoding.Unicode.GetBytes(e.Result))), Equation)
36
37            ' display left side of equation
38            questionLabel.Text = currentEquation.LeftHandSide
39            okButton.Enabled = True ' enable okButton
40            answerTextBox.Enabled = True ' enable answerTextBox
41         End If
42      End Sub ' service_DownloadStringCompleted
43
```

Fig. 23.31 | Math tutor using JSON version of `EquationGeneratorServiceJSON`. (Part 2 of 7.)

```
44      ' check user's answer
45      Private Sub okButton_Click(ByVal sender As System.Object,
46         ByVal e As System.EventArgs) Handles okButton.Click
47
48         ' check if answer field is filled
49         If Not String.IsNullOrEmpty(answerTextBox.Text) Then
50            ' determine whether user's answer is correct
51            If currentEquation.Result =
52               Convert.ToInt32(answerTextBox.Text) Then
53
54               questionLabel.Text = String.Empty ' clear question
55               answerTextBox.Clear() ' clear answer
56               okButton.Enabled = False ' disable OK button
57               MessageBox.Show("Correct! Good job!")
58            Else
59               MessageBox.Show("Incorrect. Try again.")
60            End If
61         End If
62      End Sub ' okButton_Click
63
```

Fig. 23.31 | Math tutor using JSON version of `EquationGeneratorServiceJSON`. (Part 3 of 7.)

```
64      ' set the operation to addition
65      Private Sub additionRadioButton_CheckedChanged(
66         ByVal sender As System.Object, ByVal e As System.EventArgs) _
67         Handles additionRadioButton.CheckedChanged
68         operation = "add"
69      End Sub ' additionRadioButton_CheckedChanged
70
71      ' set the operation to subtraction
72      Private Sub subtractionRadioButton_CheckedChanged(
73         ByVal sender As System.Object, ByVal e As System.EventArgs) _
74         Handles subtractionRadioButton.CheckedChanged

75         operation = "subtract"
76      End Sub ' subtractionRadioButton_CheckedChanged
77
78      ' set the operation to multiplication
79      Private Sub multiplicationRadioButton_CheckedChanged(
80         ByVal sender As System.Object, ByVal e As System.EventArgs) _
81         Handles multiplicationRadioButton.CheckedChanged
82         operation = "multiply"
83      End Sub ' multiplicationRadioButton_CheckedChanged
84
```

Fig. 23.31 | Math tutor using JSON version of `EquationGeneratorServiceJSON`. (Part 4 of 7.)

```
85      ' set difficulty level to 1
86      Private Sub levelOneRadioButton_CheckedChanged(
87         ByVal sender As System.Object, ByVal e As System.EventArgs) _
88         Handles levelOneRadioButton.CheckedChanged
89         level = 1
90      End Sub ' levelOneRadioButton_CheckedChanged
91
92      ' set difficulty level to 2
93      Private Sub levelTwoRadioButton_CheckedChanged(
94         ByVal sender As System.Object, ByVal e As System.EventArgs) _
95         Handles levelTwoRadioButton.CheckedChanged
96         level = 2
97      End Sub ' levelTwoRadioButton_CheckedChanged
98
99      ' set difficulty level to 3
100     Private Sub levelThreeRadioButton_CheckedChanged(
101        ByVal sender As System.Object, ByVal e As System.EventArgs) _
102        Handles levelThreeRadioButton.CheckedChanged
103        level = 3
104     End Sub ' levelThreeRadioButton_CheckedChanged
105  End Class ' MathTutor
```

Fig. 23.31 | Math tutor using JSON version of `EquationGeneratorServiceJSON`. (Part 5 of 7.)

a) Generating a level 2 multiplication equation.



b) Answering the question incorrectly.





Fig. 23.31 | Math tutor using JSON version of `EquationGeneratorServiceJSON`. (Part 6 of 7.)

c) Answering the question correctly.



Fig. 23.31 | Math tutor using JSON version of `EquationGeneratorServiceJSON`. (Part 7 of 7.)

```
1  ' Fig. 23.32: Equation.vb
2  ' Equation class representing a JSON object.
3  <Serializable()>
4  Public Class Equation
5     Public Left As Integer
6     Public LeftHandSide As String
7     Public Operation As String
8     Public Result As Integer
9     Public Right As Integer
10    Public RightHandSide As String
11 End Class ' Equation
```

Fig. 23.32 | Equation class representing a JSON object.



Fig. 23.33 | Template web form for phone book client.